In[ ]:=
```
Quit[]
```

In[1]:=
```
ClearAll["Global`*"]
```

Coordinate inversion for SinhSymTP coordinates

Author: Thiago Assumpção

In[2]:=
```
(*
 * We begin by defining the Cartesian coordinates (X, Y, Z)
 * in terms of the reference metric coordinates (x0, x1, x2)
 *)
X = AMAX * Sinh[x0 / SINHWAA] / Sinh[1 / SINHWAA] * Sin[x1] * Cos[x2];
Y = AMAX * Sinh[x0 / SINHWAA] / Sinh[1 / SINHWAA] * Sin[x1] * Sin[x2];
Z = ((AMAX * Sinh[x0 / SINHWAA] / Sinh[1 / SINHWAA])^2 + bScale^2)^(1/2) * Cos[x1];

sol = Solve[{X == x, Y == y, Z == z}, {x0, x1, x2}];

inversionRules = FullSimplify[sol,
   Assumptions → {AMAX > 0, bScale > 0, SINHWAA > 0, x ∈ Reals, y ∈ Reals, z ∈ Reals}];

(*
 * In the positive quadrant, e.g. x=y=z=1, we know what,
 * for any choice of bScale, SINHWAA, and AMAX,
 * the coordinates (x0, x1, x2) are all positive.
 * Hence, of all inversion rules found above, we select
 * only the ones that satisfy this condition.
 *)

testSub = {x → 1, y → 1, z → 1, bScale → 5, SINHWAA → 0.2, AMAX → 100};
inversionRulesTest = inversionRules /. testSub // N;

goodIndices = Select[Range[Length[inversionRules]],
   Module[{r}, r = inversionRules[[#]] /. testSub // N;
     AllTrue[{x0, x1, x2} /. r, Positive]] &];

(*
 * After applying the selection rule, we should
 * end up with two coordinate transformations
 * that satisfy the constraint above. Note that
 * they have the exact same transformation for
 * x0 and x1. For the coordinate x2, I recommend
 * using atan2(y, x), which already selects
 * the correct sign.
 *)
originalRulesCorresponding = inversionRules[[goodIndices]];
```

Solve : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

In[11]:=
```
x0 /. originalRulesCorresponding[[1]]
x1 /. originalRulesCorresponding[[1]]
```

Out[11]=

SINHWAA

$$\text{ArcCsch}\left[\frac{\sqrt{2}\ \text{AMAX}\ \text{Csch}\left[\frac{1}{\text{SINHWAA}}\right]}{\sqrt{-\text{bScale}^2 + x^2 + y^2 + z^2 + \sqrt{\left(x^2 + y^2 + (\text{bScale} - z)^2\right)\left(x^2 + y^2 + (\text{bScale} + z)^2\right)}}}\right]$$

Out[12]=

$$\text{ArcCos}\left[\frac{\sqrt{2}\ z}{\sqrt{\text{bScale}^2 + x^2 + y^2 + z^2 + \sqrt{\left(x^2 + y^2 + (\text{bScale} - z)^2\right)\left(x^2 + y^2 + (\text{bScale} + z)^2\right)}}}\right]$$